In[1]:=
```
"----------------------------------------------------------------------";
"PORTALE SEMPLICE INCASTRATO AL PIEDE
 File Mathematica con procedimento di condensazione statica

 Corso di Fondamenti di Dinamica e Instabilita' delle Strutture
 Universita' di Bergamo, Facolta' di Ingegneria, Dalmine
 prof. Egidio Rizzi
 Marzo 2007";
"----------------------------------------------------------------------";

"Istruzioni d'uso:
 Ogni cella di comandi puo' essere eseguita in Mathematica
 cliccando col mouse nello spazio all'interno dei delimitatori
 visibili a destra e agendo sulla tastiera con sfhit+enter";

"Disabilita la segnalazione di spelling errors";
Off[General::spell]
Off[General::spell1]

"Soluzione col Metodo degli Spostamenti";

"Gradi di liberta' q={q1,q2,q3}:
  q1=u    spostamento orizzontale del traverso
  q2=phi1 rotazione del nodo trave/pilastro di sinistra
  q3=phi2 rotazione del nodo trave/pilastro di destra";

"Forze nodali equivalenti ai carichi";
F = {F1, F2 lc, F3 lc};

"Matrice di rigidezza completa";
K11 = 24 EJc / lc^3;
K21 = 6 EJc / lc^2;
K31 = K21;
K12 = K21;
K22 = 4 EJc / lc + 4 EJt / lt;
K32 = 2 EJt / lt;
K13 = K31;
K23 = K32;
K33 = K22;
K = {{K11, K12, K13},
     {K21, K22, K23},
     {K31, K32, K33}};

"Mostra K";
MatrixForm[K]

"Soluzione del sistema lineare completo di equilibrio K.q=F";
qst = Simplify[LinearSolve[K, F]]
```

Out[24]//MatrixForm=

$$\begin{pmatrix} \frac{24\,EJc}{lc^3} & \frac{6\,EJc}{lc^2} & \frac{6\,EJc}{lc^2} \\ \frac{6\,EJc}{lc^2} & \frac{4\,EJc}{lc}+\frac{4\,EJt}{lt} & \frac{2\,EJt}{lt} \\ \frac{6\,EJc}{lc^2} & \frac{2\,EJt}{lt} & \frac{4\,EJc}{lc}+\frac{4\,EJt}{lt} \end{pmatrix}$$

Out[26]=

$$\left\{\frac{lc^3\,(3\,EJt\,F1\,lc+EJc\,(2\,F1-3\,(F2+F3))\,lt)}{12\,EJc\,(6\,EJt\,lc+EJc\,lt)},\right.$$
$$\frac{lc^2\,lt\,(-EJt\,(F1-8\,F2+4\,F3)\,lc+EJc\,(-2\,F1+5\,F2+3\,F3)\,lt)}{24\,EJt^2\,lc^2+52\,EJc\,EJt\,lc\,lt+8\,EJc^2\,lt^2},$$
$$\left.-\frac{lc^2\,lt\,(EJt\,(F1+4\,F2-8\,F3)\,lc+EJc\,(2\,F1-3\,F2-5\,F3)\,lt)}{24\,EJt^2\,lc^2+52\,EJc\,EJt\,lc\,lt+8\,EJc^2\,lt^2}\right\}$$

In[27]:=

```
"";
"Condensazione statica dei gdl statici (rotazioni)";
Kdd = {K[[1, 1]]};
Kds = {K[[1, 2]], K[[1, 3]]};
Ksd = {K[[2, 1]],
      K[[3, 1]]};
Kss = {{K[[2, 2]], K[[2, 3]]},
       {K[[3, 2]], K[[3, 3]]}};
KssI = Simplify[Inverse[Kss]];

Fd = {F[[1]]};
Fs = {F[[2]], F[[3]]};

"Rigidezza e forza condensate relative al gdl traslazionale q1=u";
Kcond = Simplify[Kdd - Kds. KssI.Ksd];
Fcond = Simplify[Fd - Kds.KssI.Fs];

"NB. In questo caso si tratta di scalari";
Kcond11 = Kcond[[1]]
Fcond11 = Fcond[[1]]
```

Out[40]=

$$\frac{12\,EJc\,(6\,EJt\,lc+EJc\,lt)}{lc^3\,(3\,EJt\,lc+2\,EJc\,lt)}$$

Out[41]=

$$\frac{3\,EJt\,F1\,lc+EJc\,(2\,F1-3\,(F2+F3))\,lt}{3\,EJt\,lc+2\,EJc\,lt}$$

In[42]:=
```
"";
"Soluzione e sostituzione a ritroso nei gdl statici";
qdst = Simplify[Kcond11^(-1) Fcond11]
qsst = Simplify[KssI.Fs - KssI.Ksd qdst]

"Verifica soluzione da condensazione statica con soluzione completa precedente";
checkq = Simplify[{qdst, qsst[[1]], qsst[[2]]} - qst]

"Azione interne all'incastro al piede destro
  Npd (positiva verso l'alto),
  Tpd (positiva verso sinistra),
  Mpd (positiva antioraria)";
Npd = Simplify[-6 EJt / lt^2 qsst[[1]] - 6 EJt / lt^2 qsst[[2]]];
Tpd = Simplify[12 EJc / lc^3 qdst + 6 EJc / lc^2 qsst[[2]]];
Mpd = Simplify[6 EJc / lc^2 qdst + 2 EJc / lc qsst[[2]]];
NTMpd = {Npd, Tpd, Mpd}
```

Out[44]=
$$\frac{lc^3\ (3\ EJt\ F1\ lc + EJc\ (2\ F1 - 3\ (F2 + F3))\ lt)}{12\ EJc\ (6\ EJt\ lc + EJc\ lt)}$$

Out[45]=
$$\left\{\frac{lc^2\ lt\ (-EJt\ (F1 - 8\ F2 + 4\ F3)\ lc + EJc\ (-2\ F1 + 5\ F2 + 3\ F3)\ lt)}{24\ EJt^2\ lc^2 + 52\ EJc\ EJt\ lc\ lt + 8\ EJc^2\ lt^2},\right.$$
$$\left.\frac{lc^2\ lt\ (-EJt\ (F1 + 4\ F2 - 8\ F3)\ lc + EJc\ (-2\ F1 + 3\ F2 + 5\ F3)\ lt)}{24\ EJt^2\ lc^2 + 52\ EJc\ EJt\ lc\ lt + 8\ EJc^2\ lt^2}\right\}$$

Out[47]=
$$\{0, 0, 0\}$$

Out[52]=
$$\left\{\frac{3\ EJt\ (F1 - 2\ (F2 + F3))\ lc^2}{lt\ (6\ EJt\ lc + EJc\ lt)}, \frac{EJt\ F1\ lc + EJc\ (2\ F1 - 3\ F2 + 3\ F3)\ lt}{2\ EJt\ lc + 4\ EJc\ lt},\right.$$
$$\left.\frac{lc\ (3\ EJt^2\ F1\ lc^2 + EJc\ EJt\ (7\ F1 - 7\ F2 + 5\ F3)\ lc\ lt + EJc^2\ (2\ F1 - 3\ F2 - F3)\ lt^2)}{12\ EJt^2\ lc^2 + 26\ EJc\ EJt\ lc\ lt + 4\ EJc^2\ lt^2}\right\}$$

In[53]:=
```
"";
"Introduzione del rapporto di rigidezza trave/
   colonna rho=Sommatoria (EJt/lt) / Sommatoria (EJc/lc)";
EJt = rho 2 EJc lt / lc;
Kcondrho = Simplify[Kcond11]
Plot[Kcondrho / (6 EJc / lc^3), {rho, 0, 10},
 PlotRange → {{0, 10}, {1, 4}}, AxesLabel → {rho, k}]
```

Out[56]=
$$\frac{6\ EJc\ (1 + 12\ rho)}{lc^3\ (1 + 3\ rho)}$$



Out[57]= - Graphics -

In[58]:=
```
"";
"Alcuni valori caratteristici del coefficiente k che premoltiplica 6 EJc/lc^3";
k = Kcondrho / (6 EJc / lc^3);
{k /. {rho → 0},
 k /. {rho → 1 / 4},
 k /. {rho → 1 / 2},
 k /. {rho → 1},
 k /. {rho → 2},
 k /. {rho → 4},
 k /. {rho → 6},
 k /. {rho → 8},
 k /. {rho → 10},
 Limit[k , rho → Infinity]}
```

Out[61]= $\left\{1, \frac{16}{7}, \frac{14}{5}, \frac{13}{4}, \frac{25}{7}, \frac{49}{13}, \frac{73}{19}, \frac{97}{25}, \frac{121}{31}, 4\right\}$

In[62]:=
```
"";
"Soluzione in funzione di rho, con casi estremi (rho=0 e rho->infinito)";
NTMpdrho = Simplify[NTMpd]
NTMpd0 = Simplify[NTMpd /. {rho → 0}]
NTMpdinf = Simplify[Limit[NTMpd, rho → Infinity]]

qstrho = Simplify[{qdst, qsst[[1]], qsst[[2]]}]
qst0 = Simplify[{qdst, qsst[[1]], qsst[[2]]} /. {rho → 0}]
qstinf = Simplify[Limit[{qdst, qsst[[1]], qsst[[2]]}, rho → Infinity]]
```

Out[64]= $\left\{\frac{6\,(F1 - 2\,(F2 + F3))\,lc\,rho}{lt + 12\,lt\,rho}, \frac{2\,F1 - 3\,F2 + 3\,F3 + 2\,F1\,rho}{4 + 4\,rho}, \frac{lc\,(-3\,F2 - F3 - 14\,F2\,rho + 10\,F3\,rho + 2\,F1\,(1 + 7\,rho + 6\,rho^2))}{4 + 52\,rho + 48\,rho^2}\right\}$

Out[65]= $\left\{0, \frac{1}{4}\,(2\,F1 - 3\,F2 + 3\,F3), \frac{1}{4}\,(2\,F1 - 3\,F2 - F3)\,lc\right\}$

Out[66]= $\left\{\frac{(F1 - 2\,(F2 + F3))\,lc}{2\,lt}, \frac{F1}{2}, \frac{F1\,lc}{4}\right\}$

Out[67]= $\left\{\frac{lc^3\,(-3\,(F2 + F3) + F1\,(2 + 6\,rho))}{12\,(EJc + 12\,EJc\,rho)}, -\frac{lc^2\,(2\,F1\,(1 + rho) + F3\,(-3 + 8\,rho) - F2\,(5 + 16\,rho))}{8\,EJc\,(1 + 13\,rho + 12\,rho^2)}, -\frac{lc^2\,(2\,F1\,(1 + rho) + F2\,(-3 + 8\,rho) - F3\,(5 + 16\,rho))}{8\,EJc\,(1 + 13\,rho + 12\,rho^2)}\right\}$

Out[68]= $\left\{\frac{(2\,F1 - 3\,(F2 + F3))\,lc^3}{12\,EJc}, \frac{(-2\,F1 + 5\,F2 + 3\,F3)\,lc^2}{8\,EJc}, \frac{(-2\,F1 + 3\,F2 + 5\,F3)\,lc^2}{8\,EJc}\right\}$

Out[69]= $\left\{\frac{F1\,lc^3}{24\,EJc}, 0, 0\right\}$

In[70]:=
```
"";
"Soluzione alternativa col Metodo delle Forze tramite il PLV";

"Scelta delle tre incognite iperstatiche:
 X1=Npd (positiva verso l'alto),
 X2=Tpd (positiva verso sinistra),
 X3=Mpd (positiva antioraria)";

"Momento nella struttura principale isostatica";
M0a = 0;
M0b = F3 lc;
```

```
M0c = (F2 + F3) lc - F1 x;

"Momenti nella tre strutture fittizie";
M1a = 0;
M1b = x;
M1c = lt;

M2a = -x;
M2b = -lc;
M2c = -lc + x;

M3a = 1;
M3b = 1;
M3c = 1;

"Calcolo degli integrali da PLV";

A11 = Simplify[Integrate[M1a M1a / EJc + M1c M1c / EJc, {x, 0, lc}] +
    Integrate[M1b M1b / EJt, {x, 0, lt}]];
A12 = Simplify[Integrate[M1a M2a / EJc + M1c M2c / EJc, {x, 0, lc}] +
    Integrate[M1b M2b / EJt, {x, 0, lt}]];
A13 = Simplify[Integrate[M1a M3a / EJc + M1c M3c / EJc, {x, 0, lc}] +
    Integrate[M1b M3b / EJt, {x, 0, lt}]];
A22 = Simplify[Integrate[M2a M2a / EJc + M2c M2c / EJc, {x, 0, lc}] +
    Integrate[M2b M2b / EJt, {x, 0, lt}]];
A23 = Simplify[Integrate[M2a M3a / EJc + M2c M3c / EJc, {x, 0, lc}] +
    Integrate[M2b M3b / EJt, {x, 0, lt}]];
A33 = Simplify[Integrate[M3a M3a / EJc + M3c M3c / EJc, {x, 0, lc}] +
    Integrate[M3b M3b / EJt, {x, 0, lt}]];

a10 = Simplify[-Integrate[M1a M0a / EJc + M1c M0c / EJc, {x, 0, lc}] -
    Integrate[M1b M0b / EJt, {x, 0, lt}]];
a20 = Simplify[-Integrate[M2a M0a / EJc + M2c M0c / EJc, {x, 0, lc}] -
    Integrate[M2b M0b / EJt, {x, 0, lt}]];
a30 = Simplify[-Integrate[M3a M0a / EJc + M3c M0c / EJc, {x, 0, lc}] -
    Integrate[M3b M0b / EJt, {x, 0, lt}]];

A = {{A11, A12, A13},
    {A12, A22, A23},
    {A13, A23, A33}};
a = {a10,
    a20,
    a30};

"Vettore incognite iperstatiche in soluzione";
X = Simplify[LinearSolve[A, a]]

"Controllo delle soluzioni statiche
   coi due metodi degli spostamenti e delle forze";
checkNTMpd = Simplify[NTMpd - X]
```

Out[100]=

$$\left\{\frac{6\,(F1-2\,(F2+F3))\,lc\,rho}{lt+12\,lt\,rho},\ \frac{2\,F1-3\,F2+3\,F3+2\,F1\,rho}{4+4\,rho},\ \frac{lc\,(-3\,F2-F3-14\,F2\,rho+10\,F3\,rho+2\,F1\,(1+7\,rho+6\,rho^2))}{4+52\,rho+48\,rho^2}\right\}$$

Out[102]=

$$\{0,\ 0,\ 0\}$$

In[103]:=

```
"";
"Calcolo spostamenti col PLV";
Ma = M0a + X[[1]] M1a + X[[2]] M2a + X[[3]] M3a;
Mb = M0b + X[[1]] M1b + X[[2]] M2b + X[[3]] M3b;
Mc = M0c + X[[1]] M1c + X[[2]] M2c + X[[3]] M3c;

u = Simplify[Integrate[(-x) Mc / EJc, {x, 0, lc}]];
phi1 = Simplify[Integrate[(1) Mc / EJc, {x, 0, lc}]];
phi2 =
  Simplify[Integrate[(1) Mc / EJc, {x, 0, lc}] + Integrate[(1) Mb / EJt, {x, 0, lt}]];

qstPLV = {u, phi1, phi2}

"Controllo delle soluzioni cinematiche
   coi due metodi degli spostamenti e delle forze";
checkqst = Simplify[qst - qstPLV]
```

Out[111]=

$$\left\{\frac{lc^3\,(-3\,(F2+F3)+F1\,(2+6\,rho))}{12\,(EJc+12\,EJc\,rho)},\ \frac{lc^2\,(F3\,(3-8\,rho)-2\,F1\,(1+rho)+F2\,(5+16\,rho))}{8\,EJc\,(1+13\,rho+12\,rho^2)},\ \frac{lc^2\,(F2\,(3-8\,rho)-2\,F1\,(1+rho)+F3\,(5+16\,rho))}{8\,EJc\,(1+13\,rho+12\,rho^2)}\right\}$$

Out[113]=

$$\{0,\ 0,\ 0\}$$

In[114]:=

```
"";
"Condizione di carico con la sola forza orizzontale F1 sul traverso";

NTMpdF1 = Simplify[NTMpd /. {F2 → 0, F3 → 0}]
qstF1 = Simplify[qst /. {F2 → 0, F3 → 0}]

"Andamento dello spostamento orizzontale
   u=f F lc^3/EJc del traverso in funzione di rho";
Plot[qstF1[[1]] * (EJc / lc^3 / F1), {rho, 0, 10}, PlotRange → {{0, 10}, {0.04, 0.18}},
  AxesLabel → {rho, "f=1/(6k)"}, AxesOrigin → {0, 0.03999}]

"Caso del traverso infinitamente flessibile (rho = 0) ed
 infinitamente rigido (rho -> infinito)";
Simplify[NTMpdF1 /. {rho → 0}]
Simplify[qstF1 /. {rho → 0}]

Simplify[Limit[NTMpdF1 , rho → Infinity]]
Simplify[Limit[qstF1, rho → Infinity]]
```

Out[116]=

$$\left\{\frac{6\,F1\,lc\,rho}{lt+12\,lt\,rho},\ \frac{F1}{2},\ \frac{lc\,(F1+6\,F1\,rho)}{2+24\,rho}\right\}$$

Out[117]=

$$\left\{\frac{lc^3\,(F1+3\,F1\,rho)}{6\,(EJc+12\,EJc\,rho)},\ \frac{F1\,lc^2}{-4\,EJc-48\,EJc\,rho},\ \frac{F1\,lc^2}{-4\,EJc-48\,EJc\,rho}\right\}$$



Out[119]=

- Graphics -

Out[121]=

$$\left\{0,\ \frac{F1}{2},\ \frac{F1\,lc}{2}\right\}$$

Out[122]=

$$\left\{\frac{F1\,lc^3}{6\,EJc},\ -\frac{F1\,lc^2}{4\,EJc},\ -\frac{F1\,lc^2}{4\,EJc}\right\}$$

Out[123]=

$$\left\{\frac{F1\,lc}{2\,lt},\ \frac{F1}{2},\ \frac{F1\,lc}{4}\right\}$$

Out[124]=

$$\left\{\frac{F1\,lc^3}{24\,EJc},\ 0,\ 0\right\}$$

```
"";
"Una eventuale ulteriore soluzione alternativa col Metodo delle Forze
 potrebbe essere sviluppata di seguito mediante il Metodo della Linea Elastica";
"";
```